DataReader Asíncrono en Visual Basic 2005
A veces nos encontramos con procesos que toman mucho tiempo, por ejemplo accesando datos.

Cuando esto sucede, la interfaz del usuario puede quedar “congelada” mientras dure el proceso.

Queremos evitar este “congelamiento”.

Otra necesidad es cuando tenemos que llevar a cabo varios procesos, y cada proceso toma su tiempo.

Si llamamos a los procesos uno por uno, en serie, el tiempo se acumula.

Si pudieramos llamar a los procesos en paralelo, se ahorraría tiempo.

Usaremos procesos “Asíncronos”, en los cuales llamamos el proceso, pero NO nos quedamos esperando a recibir la respuesta. Sino que el proceso nos avisará cuando haya terminado. Específicamente, al terminar, se corre una rutina para obtener los resultados.
Para este ejemplo, tendremos dos proyectos: 
Un “cliente” Windows Forms (winAppAsyncSQL) y una librería o DLL (libDataStore):

Nota: yo acostumbro eliminar la “clase” que viene en el proyecto tipo Librería, y adicionar un “componente” 

(en nuestro caso, el componente: compDataStore.vb). Los componentes pueden tener mayor funcionalidad que las “clases” simples.

Código del Componente compDataStore
Public Class compDataStore

#Region "Popular Datos Asincrónicamente"

    Public Function PopularDatosAsincronicamente( _

        ByVal sSQL As String, _

        ByVal callBack As AsyncCallback) As IAsyncResult
        Try

            '----------------------------------------------------------------

            ' Connection String Builder 

            ' (para armar Connection String)

            '----------------------------------------------------------------

            Dim csbCnn As New SqlClient.SqlConnectionStringBuilder

            csbCnn.DataSource = "(local)"

            csbCnn.InitialCatalog = "Northwind"

            csbCnn.IntegratedSecurity = True

            csbCnn.AsynchronousProcessing = True ' Para que sea asíncrono

            '----------------------------------------------------------------

            ' Connection

            '----------------------------------------------------------------

            Dim Cnn As New SqlClient.SqlConnection( _

                csbCnn.ConnectionString)

            '----------------------------------------------------------------

            ' Command

            '----------------------------------------------------------------

            Dim cmd As New SqlClient.SqlCommand(sSQL, Cnn)

            '----------------------------------------------------------------

            ' Declara Resultado Asíncrono (Nota: no lleva NEW)

            '----------------------------------------------------------------

            Dim iAR As IAsyncResult = Nothing

            '----------------------------------------------------------------

            ' Usa la Conexión e invoca Comando

            '----------------------------------------------------------------

            Cnn.Open()

            '----------------------------------------------------------------

            ' Resultado Asíncrono devuelto al ejecutar Comando

            ' Los parámetros son: Callback, State Object, Behaviour

            '----------------------------------------------------------------

            iAR = cmd.BeginExecuteReader( _

               callBack, _

               cmd, _

               CommandBehavior.CloseConnection)

            '----------------------------------------------------------------

            ' Retorna Resultado Asíncrono

            '----------------------------------------------------------------

            Return iAR

        Catch ex As Exception

            Throw ex

        End Try

    End Function

#End Region

End Class

Código del Componente winAppAsyncSQL
Public Class frmAsyncSQL
    ' Basado en Articulo en:

    ' http://www.devx.com/dotnet/Article/22358


#Region "Declaraciones"

    ' Delegates (Apuntadores)

    ' a Rutinas que actualizan el User Interface (UI) desde el Thread principal del formulario

    ' (Pues NO debe actualizar UI desde otros threads)

    Private Delegate Sub delegBindData(ByVal dt As DataTable)

    Private Delegate Sub delegErrorMessage(ByVal message As String)

    ' Para saber si ya hay un proceso corriendo (y no volverlo a llamar)

    Private isAsync As Boolean = False

#End Region


#Region "Asíncrono"

    Private Sub btnCargarAsincrono_Click( _

        ByVal sender As System.Object, _

        ByVal e As System.EventArgs) Handles btnCargarAsincrono.Click

        Try

            Me.tsslStatus.Text = "Procesando..."

            Application.DoEvents()

            If Not isAsync Then

                '----------------------------------------------------------------

                ' Instancia Componente de Acceso a Datos

                '----------------------------------------------------------------

                Dim objDataStore As New libDataStore.compDataStore

                '----------------------------------------------------------------

                ' Lee Clientes usando el Componente

                '----------------------------------------------------------------

                Dim sSQLCustomers As String

                sSQLCustomers = "SELECT * FROM Customers"

                Dim iAsyncResultCustomers As IAsyncResult

                iAsyncResultCustomers = _

                    objDataStore.PopularDatosAsincronicamente( _

                    sSQLCustomers, AddressOf procesarClientes)

                Me.tsslStatus.Text = "Procesando... (llamada hecha)"

                Application.DoEvents()

                MessageBox.Show("Llamada Hecha")

            End If

        Catch ex As Exception

            MessageBox.Show(ex.Message)

        End Try

    End Sub

#End Region


#Region "Procesar Clientes (Customers)"

    Private Sub procesarClientes( _

        ByVal iAsyncResultCustomers As IAsyncResult)

        Try

            '----------------------------------------------------------------

            ' Flag de que está procesando

            '----------------------------------------------------------------

            isAsync = True

            '----------------------------------------------------------------

            ' Data Table

            '----------------------------------------------------------------

            Dim dt As New DataTable

            '----------------------------------------------------------------

            ' Siempre chequear, que realmente esté completo

            '----------------------------------------------------------------

            While iAsyncResultCustomers.IsCompleted

                '----------------------------------------------------------------

                ' En el State Object guardamos el Comando SQL...

                ' Volvemos a recobrar el Comando SQL a través del State Object

                '----------------------------------------------------------------

                Dim cmd As SqlClient.SqlCommand

                cmd = CType( _

                    iAsyncResultCustomers.AsyncState, _

                    SqlClient.SqlCommand)



                '----------------------------------------------------------------

                ' Obtenemos el Data Reader

                ' a través del Resultado Asíncrono

                '----------------------------------------------------------------

                Dim rd As SqlClient.SqlDataReader

                rd = cmd.EndExecuteReader(iAsyncResultCustomers)

                ' Using llama al Dispose de rd al final

                ' con lo que ayuda a limpiar...

                Using rd

                    Try

                        ' Carga los datos del Reader al DataTable!

                        dt.Load(rd)

                        ' Llama a rutinas que actualizan UI...

                        ' Recordar: NO se debe actualizar UI

                        ' estando en otro thread que no sea el Main thread

                        Dim bindDataCallBack As delegBindData

                        bindDataCallBack = New delegBindData(AddressOf bindCustomers)

                        ' bindDataCallBack es un Delegate

                        ' dt es un arreglo de parámetros a pasar en la invocación

                        Me.Invoke(bindDataCallBack, dt)

                    Catch ex As Exception

                        ' Llama a rutina que muestra error

                        Dim errorMessageCallBack As delegErrorMessage

                        errorMessageCallBack = New delegErrorMessage(AddressOf showError)

                        Me.Invoke(errorMessageCallBack, ex.Message)

                    End Try

                End Using

            End While

            isAsync = False

        Catch ex As Exception

            isAsync = False

            Throw ex

        End Try

    End Sub

#End Region

#Region "Actualizar UI Clientes (Customers)"


    Private Sub bindCustomers(ByVal dt As DataTable)

        Try

            Me.dgvClientes.DataSource = dt

            Me.tsslStatus.Text = "Datos Cargados!!!"

        Catch ex As Exception

            MessageBox.Show(ex.Message)

        End Try

    End Sub


    Private Sub showError(ByVal message As String)

        Try

            MessageBox.Show("Error: " & message)

        Catch ex As Exception

            MessageBox.Show(ex.Message)

        End Try

    End Sub

#End Region

End Class

